Zakrivljeni prostori

Inicijalizacija

```
In[1]:= << Graphics`SurfaceOfRevolution`

In[2]:= SetOptions[{Plot3D, ParametricPlot3D, SurfaceOfRevolution},
          Axes → False, Boxed → False];
        SetOptions[Plot3D, DisplayFunction → Identity];
```

Zakrivljenost plohe u točki

Zakrivljenost κ(P) *krivulje C* u nekoj točki P je inverz radijusa kružnice koja "paše" u krivulju u okolini te točke. (Druge derivacije se podudaraju.)

```
In[4]:= Show[Import["zakrivljenost.gif"]]
```




```
Out[4]= - Graphics -
```

Zakrivljenost *k(P) plohe* $\mathcal{M}$ u točki P definirana je na slijedeći način. Presjecimo plohu ravninom $\mathcal{R}$ koja sadrži točku P i vektor **n** normalan na plohu u toj točki. Neka je sada κ zakrivljenost krivulje presjeka $\mathcal{M}$ i $\mathcal{R}$, gdje je predznak od κ definiran prema tome da li je centar zakrivljenosti u smjeru normale **n** ili suprotnom. Tzv. *principalne zakrivljenosti* $\kappa_{\min}$ i $\kappa_{\max}$ su minimalna i maksimalna vrijednost zakrivljenosti κ, kad $\mathcal{R}$ rotira oko **n**. Zakrivljenost k(p) (*Gaussova zakrivljenost, intrinsična zakrivljenost*) je onda definirana kao

$$k = \kappa_{\min}\ \kappa_{\max}\ .$$

(Usput, Gauss je dokazao da su smjerovi ovih dvaju principalnih zakrivljenosti međusobno okomiti.). Nas će ovdje zanimati prvenstveno predznak od k tj. razlika između pozitivno zakrivljenih, negativno zakrivljenih i ravnih ploha.

Nacrtajmo nekoliko ploha:

```
In[5]:= pozplot =
          Plot3D[Sin[x] Cos[y], {x, 0, π}, {y, -π/3, π/3}, ViewPoint -> {1.944, -2.744, 0.379}];
        negplot = Plot3D[Sin[x] Cosh[y], {x, 0, π}, {y, -π, π},
           ViewPoint -> {2.340, -1.795, 1.659}];
        ravplot1 = Plot3D[(-(x - 2)^2 + 4), {x, 0, 4}, {y, -π/2, π/2},
           ViewPoint -> {2.340, -2.795, 1.659}];
        ravplot2 = Plot3D[1, {x, 0, 4}, {y, -π/2, π/2}, ViewPoint -> {2.340, -1.795, 1.659}];
        Show[GraphicsArray[{{negplot, pozplot}, {ravplot2, ravplot1}}],
         GraphicsSpacing → {0, 0}]
```



```
Out[9]= - GraphicsArray -
```

U tipičnim točkama ovih ploha zakrivljenost je negativna (za plohu br. 1), pozitivna (za plohu br. 2) i nula (za plohe br. 3 i 4). Zakrivljenost prvih dviju ploha nije konstantna tj. ovisi o točki koju gledamo. Ova zakrivljenost je intrinsično svojstvo plohe i ne mijenja se ako plohu "savijamo", što se zorno vidi na plohama br. 3 i 4 koje su obje intrinsično ravne. Ploha br. 4 je zakrivljena "ekstrinsično" tj. u trećoj dimenziji koja je izvan plohe. Dvodimenzionalna bića nikakvim mjerenjima ograničenima na plohu ne bi mogla razlikovati plohe br. 3 i 4. Jedno važno svojstvo intrinsično zakrivljene plohe je da se vektor kojeg paralelno transportiramo po zatvorenoj krivulji ne mora vratiti u ishodišni položaj jednako usmjeren kao na početku.

Plohe konstantne pozitivne zakrivljenosti

Dvodimenzionalna ploha *konstantne* pozitivne zakrivljenosti zove se *2−sfera*, i može se definirati jednadžbom

$$x_1^{\,2} + x_2^{\,2} + x_3^{\,2} = S^2$$

gdje je S polumjer sfere. To je obična sfera iz svakodnevnog života, npr. površina Zemljine kugle. Prema gornjoj definiciji njena zakrivljenost je $k = 1/S^2$.

Važno je uočiti dvije stvari:

1) Jedna od tri koordinate $x_{1,2,3}$ nije neovisna, npr. $x_3 = \sqrt{S^2 - x_1{}^2 - x_2{}^2}$ i uveli smo je samo zato da sferu prikažemo kao plohu uklopljenu u 3D prostor. Sva zamisliva događanja i mjerenja na sferi moguće je opisati pomoću samo dvije koordinate, npr uobičajene θ i ϕ.

2) Bića ograničena na plohu sfere lako mogu zaključiti da je ona intrinsično zakrivljena, npr. mjereći kuteve trokuta: njihov zbroj će uvijek biti veći od 180 stupnjeva.

Po analogiji, trodimenzionalna hiper−ploha, dakle prostor konstantne pozitivne zakrivljenosti se zove 3−sfera i zadan je jednadž bom

$$x_1{}^2 + x_2{}^2 + x_3{}^2 + x_4{}^2 = S^2$$

Naš svemir, kad bi u njemu bilo dovoljno mase, bi mogao biti takav prostor.

Koordinata x_4 nema nikakve veze s vremenom i služi nam samo da bi mogli interpretirati ovu 3−sferu kao zakrivljenu plohu *uklopljenu* u euklidski četverodimenzionalni prostor. Da bismo je vizualizirali poslužimo se parametrizacijom preko polarnih koordinata**:**

```
In[10]:= x1[χ_, θ_, ϕ_] := S Sin[χ] Cos[θ]
         x2[χ_, θ_, ϕ_] := S Sin[χ] Sin[θ] Cos[ϕ]
         x3[χ_, θ_, ϕ_] := S Sin[χ] Sin[θ] Sin[ϕ]
         x4[χ_, θ_, ϕ_] := S Cos[χ]
```

Provjerimo prvo da je ova parametrizacija dobra:

```
In[14]:= x1[χ, θ, ϕ]^2 + x2[χ, θ, ϕ]^2 + x3[χ, θ, ϕ]^2 + x4[χ, θ, ϕ]^2
```

```
Out[14]= S^2 Cos[χ]^2 + S^2 Cos[θ]^2 Sin[χ]^2 + S^2 Cos[ϕ]^2 Sin[θ]^2 Sin[χ]^2 + S^2 Sin[θ]^2 Sin[ϕ]^2 Sin[χ]^2
```

```
In[15]:= Simplify[%]
```

```
Out[15]= S^2
```

Kao što i treba biti. Stavimo sada S→1 radi jednostavnosti.

```
In[16]:= S = 1;
```

Prikažimo djelomično ovu 3−sferu (uklopljenu u 4D prostor) u 3D prostoru, tako da promotrimo njen presjek s ravninom ϕ=0:

```
In[17]:= ParametricPlot3D[{x1[χ, θ, 0], x2[χ, θ, 0], x4[χ, θ, 0]} , {χ, 0, 2 π}, {θ, 0, π}]
```



```
Out[17]= - Graphics3D -
```

Dobili smo običnu 2D plohu konstantne pozitivne zakrivljenosti (2−sferu).

Plohe konstantne negativne zakrivljenosti

Slično, 3D prostor (hiper−plohu) konstantne *negativne* zakrivljenosti možemo uklopiti u 4D prostor tako da ga opišemo jednadž bom

$$x_1{}^2 + x_2{}^2 + x_3{}^2 - x_4{}^2 = -S^2$$

Minus prije $x_4{}^2$ znaci da ovaj put ne uklapamo plohu u obični euklidski vec u tzv. ***pseudo−euklidski*** prostor u kojem npr. dužine mogu imati negativnu duljinu. (To se ne vidi odmah iz ovih minusa već tek nakon drugačije analize za koju vidi literaturu.) Takav se pseudo−euklidski prostor pojavljuje i u specijalno−relativističkoj fizici gdje prostorno−vremenski intervali mogu imati negativnu "duljinu" $ds^2 = dt^2 - dr^2 < 0$. No ovdje je riječ o pomoćnom 4D prostoru i x_4 koordinata opet nema nikakve veze s vremenskom koordinatom. I ovdje se možemo poslužiti parametrizacijom u polarnim koordinatama:

```
In[18]:= Clear[S]

In[19]:= x1[χ_, θ_, ϕ_] := S Sinh[χ] Cos[θ]
         x2[χ_, θ_, ϕ_] := S Sinh[χ] Sin[θ] Cos[ϕ]
         x3[χ_, θ_, ϕ_] := S Sinh[χ] Sin[θ] Sin[ϕ]
         x4[χ_, θ_, ϕ_] := S Cosh[χ]
```

Provjera da je ova parametrizacija dobra:

```
In[23]:= x1[χ, θ, ϕ]^2 + x2[χ, θ, ϕ]^2 + x3[χ, θ, ϕ]^2 - x4[χ, θ, ϕ]^2

Out[23]= -S^2 Cosh[χ]^2 + S^2 Cos[θ]^2 Sinh[χ]^2 +
         S^2 Cos[ϕ]^2 Sin[θ]^2 Sinh[χ]^2 + S^2 Sin[θ]^2 Sin[ϕ]^2 Sinh[χ]^2

In[24]:= Simplify[%]

Out[24]= -S^2
```

Presjek ϕ=0 kroz 4D pseudo−euklidski prostor nam taj 3D prostor prikazuje kao 2D hiperboloid:

```
In[25]:= S = 1;
```

```
In[26]:= ParametricPlot3D[{x1[χ, θ, 0], x2[χ, θ, 0], x4[χ, θ, 0]},
          {χ, -π, π}, {θ, -π, π}, ViewPoint → {3.210, -0.565, 0.910}]
```



```
Out[26]= - Graphics3D -
```

Na gornjoj skici nije jasno da je originalna hiper−ploha *negativno* zakrivljena (ranija definicija daje pozitivnu zakrivljenost ove nacrtane plohe), ali to je artefakt toga što je uklapanje izvedeno u pseudo−euklidski, a mi vizualiziramo presjek u našem euklidskom prostoru. Vidimo da ovakvo oslanjanje na naš 3D prostorni zor može biti opasno i najbolje je pri određivanju zakrivljenosti služiti se egzaktnim analitičkim metodama, poput onih u slijedećem odjeljku.

Usput, ploha konstantne negativne zakrivljenosti u 3D *euklidskom* prostoru je tzv. *Beltramijeva pseudosfera* − ploha revolucije traktrise (*tractrix*) :

```
In[27]:= SurfaceOfRevolution[{1/Cosh[t], t - Tanh[t]},
           {t, 0, 5}, PlotRange->{0,2.0}, PlotPoints->25, ViewPoint->{1.598, -2.687,
         1.294}, Axes->False, Boxed->False]
```



```
Out[27]= - Graphics3D -
```

Vidimo da je svaka točka sedlena i da je u svakoj točki zakrivljenost negativna.

Polumjer i opseg kružnice u zakrivljenom prostoru

Pogledajmo omjer opsega i polumjera kružnice u prostorima konstantne zakrivljenosti, dakle opisanima metrikom

$$ds^2 = \frac{dr^2}{1-\mathrm{k}r^2/S^2} + r^2\, d\Omega$$

Opseg se mjeri duž θ−koordinate i uvijek je jednak 2πR:

In[28]:= $\int_0^{2\pi} \mathbf{R}\, d\theta$

Out[28]= 2π R

dok se polumjer mjeri duž r−koordinate

$$\text{polumjer} = \int_0^R \frac{dr}{\sqrt{1-k\,r^2/S^2}}$$

pa je njihov omjer:

In[29]:= **Clear[S]; (* jer nam je gore S=1 *)**

In[30]:= **Omjer[k_] := $\dfrac{2\pi R}{\text{Integrate}\left[\frac{1}{\sqrt{1-k*r^2/S^2}}, \{r, 0, R\}, \text{Assumptions} \to \{S > R > 0\}\right]}$**

U ravnom prostoru (k=0) taj omjer je dobro poznatih 2π:

In[31]:= **Omjer[0]**

Out[31]= 2π

U pozitivno zakrivljenom prostoru (k = 1) taj omjer je uvijek manji od 2π:

In[32]:= **Omjer[1]**

Out[32]= $\dfrac{2\pi R}{S\,\text{ArcSin}\left[\frac{R}{S}\right]}$

```
In[33]:= Plot[2 π x / ArcSin[x], {x, 0, 0.95}, PlotRange → {{0, 1}, {0, 7}},
          Ticks → {{0.5, 1}, {{0, "0"}, {3.14, "π"}, {6.28, "2π"}}},
          AxesLabel → {"R/S", "opseg/polumjer"}]
```




```
Out[33]= - Graphics -
```

U negativno zakrivljenom prostoru (k = −1) omjer je uvijek veći od 2π

```
In[34]:= Omjer[-1]
```

Out[34]= $\frac{2\,\pi\,R}{S\,\mathrm{ArcSinh}\left[\frac{R}{S}\right]}$

```
In[35]:= Plot[2 π x / ArcSinh[x], {x, 0, 0.95}, PlotRange → {{0, 1}, {0, 7}},
          Ticks → {{0.5, 1}, {{0, "0"}, {3.14, "π"}, {6.28, "2π"}}},
          AxesLabel → {"R/S", "opseg/polumjer"}]
```




```
Out[35]= - Graphics -
```

U granici beskonačnog radijusa zakrivljenosti (S→∞) omjer teži k 2π.

Dakle, računajući ovaj omjer lako određujemo zakrivljenost plohe.